```
> restart; Digits:=14:
> # Solve the air drag problem. Free fall with a parachute.
> # Measure downward velocity as positive
> N2law:=diff(v(t),t)=g-C/m*v(t)^2;
```

$$N2law := \frac{d}{dt}\, v(t) = g - \frac{C\, v(t)^2}{m} \qquad (1)$$

```
> g:=10.; # we work in SI to about 2 digits
```

$$g := 10. \qquad (2)$$

```
> m:=70.; # average person's mass (kg)
```

$$m := 70. \qquad (3)$$

```
> C0:=0.5*rho*A;
```

$$C0 := 0.5\, \rho\, A \qquad (4)$$

```
> A:=25; #5 by 5 meters is 25 square meters
```

$$A := 25 \qquad (5)$$

```
> rho:=1.2; # in kg/m^3, this is from Wiki, sea level at 20 degrees
(will be less at higher altitudes)
```

$$\rho := 1.2 \qquad (6)$$

```
> C0;
```

$$15.00 \qquad (7)$$

```
> # Unit of C? [C] = Accel * Mass / Velocity^2  = Mass/Length
> # formula: [rho*A] = (Mass/Volume) * Area  =  Mass/Length

> C:=C0; #
```

$$C := 15.00 \qquad (8)$$

```
> sol:=dsolve({N2law,v(0)=0.}); # Maple can solve it
```

$$sol := v(t) = \frac{10000000}{21428571428571} \sqrt{214285714285710}\, \tanh\left(\frac{1}{10000000}\, t \sqrt{214285714285710}\right) \qquad (9)$$

```
> V:=evalf(rhs(sol));
```

$$V := 6.8313005106399 \tanh(1.4638501094228\, t) \qquad (10)$$

```
> plot(V,t=0..5,thickness=3,title="Fall velocity vs time when v(0)=
0 (in SI)");
```




```
> # is this good enough to survive a fall?
> 6.8*3.6; # in km/h - doable ? (with practice, yes)
```

$$24.48 \quad (11)$$

```
> # It took 1-2 seconds to reach terminal velocity
> # Now allow free fall for a second, then the parachute opens?
> # How big will the acceleration be once the parachute kicks in?

> # free fall for 1 second - initial set up for drag problem:
> v0:=g*1;
```

$$v0 := 10. \quad (12)$$

```
> # when the parachute kicks in v0 is 10 m/s.

> sol:=dsolve({N2law,v(0)=v0}); # Maple solves !
```

$$sol := v(t) = \frac{10000000}{21428571428571} \sqrt{214285714285710} \tanh\left(\frac{1}{214285714285710000000} \left(21428571428571\, t + 1000000 \sqrt{214285714285710} \operatorname{arctanh}\left(\frac{1}{10000000} \sqrt{214285714285710}\right)\right) \sqrt{214285714285710}\right) \quad (13)$$

```
> V:=evalf(rhs(sol));
```

$$V := 6.8313005106399 \tanh(1.4638501094228\, t + 0.83495951529394 - 1.5707963267949\, I) \quad (14)$$

```
> plot(V,t=0..5,thickness=3,title="Fall velocity vs time when v(0)=10 (in SI)");
```




```
> # The same terminal velocity is achieved!
> # Get the accel. by derivative:
> a:=diff(V,t);
```

$$a := (10.000000000000 + 0.\, I) \left(1 - \tanh(1.4638501094228\, t + 0.83495951529394 - 1.5707963267949\, I)^2\right) \quad (15)$$

```
> plot(a,t=0..2, thickness=3,title="Acceleration vs time when parachute opens (in SI)");
```




```
> # If this was realistic (immediate opening of parachute, ie, area
  A goes from 0 to full, then the parachutist would undergo a
  massive stretch: before time zero s/he accelerates with g=10
  m/s^2 downwards (positive direction by choice), and then in one
  instant the acceleration switches to the opposite, 11 m/s^2 in
  the upward direction! This would not be good for the body!
```